/* Departemental level estimates */
/* LOCATION SHIFT */
clear all
cd "C:\Users\Public\Documents\ImmPanelRevis19\"
cd "ImmPanelRevis19\Donnees\select"

/***************/
/* Estimations */
/***************/
capture program drop regsamp
program define regsamp
version 11
syntax, occp(name) outc(name)

use dep_final.dta, clear

gen reg=substr(Xze,1,2)
/*gen nreg="11" if reg=="11" | reg=="00"
replace nreg="24" if reg=="24"
replace nreg="27" if reg=="26" | reg=="43"
replace nreg="28" if reg=="23" | reg=="25"
replace nreg="32" if reg=="31" | reg=="22"
replace nreg="44" if reg=="42" | reg=="21" | reg=="41"
replace nreg="52" if reg=="52" 
replace nreg="53" if reg=="53" 
replace nreg="75" if reg=="72" | reg=="74" | reg=="54"
replace nreg="76" if reg=="91" | reg=="73"
replace nreg="84" if reg=="83" | reg=="82"
replace nreg="93" if reg=="93" 
replace nreg="94" if reg=="94" */

xi , pre(REG_) i.reg

gen occ="bc" if bc==1
replace occ="cad" if cad==1

/* adjust for ile de France mobility */
gen idf=(ze=="75" | ze=="77" | ze=="78" | ze=="91" |ze=="92" |ze=="93" |ze=="94" |ze=="95" )
gen aidf=(aze=="75" | aze=="77" | aze=="78" | aze=="91" | aze=="92" | aze=="93" | aze=="94" | aze=="95" )


/* create some outcome variables */
gen ldp=ln(dp/l.dp)
gen dftfy=(dp==360)-(L.dp==360)
gen oshift=(`occp'==0 & L.`occp'==1)
replace oshift=( (cad==0 & L.cad==1) | (tecemp2==0 & L.tecemp2==1) | (bc==0 & L.bc==1) | (bcnq==0 & L.bcnq==1)) if "`occp'"=="all"

/* keep individuals in the occupation group */
keep if (btime==1 & `occp') | (btime==2 & L.`occp')

/* location in the decile of the initial group distribution */
gen lsnd=ln(sn/dp)
/* residual wages for each year */
quietly: reg lsnd F_* 
predict rlsnd, residuals
/* change in residual wages */
gen drlsnd=D.rlsnd
/* residual outcome */
quietly: reg `outc' F_*  if btime==2
predict r`outc' if btime==2, residuals

/* rank in the wage distribution */
drop F_*
egen qinit = fastxtile(rlsnd) if ftreat, by(ze an btime) nq(100)
replace qinit=qinit/100

/* interaction term and instruments */
gen lqinit=L.qinit
gen inter= dimm*L.qinit
gen linter=ldimm*L.qinit
gen inter75= dimm75 * L.qinit
gen linter75=ldimm75*L.qinit
gen interm2= dimtm2 * L.qinit
gen linterm2=ldimtm2 *L.qinit

keep if btime==2 & L.`occp' & !missing(dimm) & !missing(dimm75)
foreach year in 1982 1991 1999 2007 {
/* winsorize drlsnd */
quietly: sum drlsnd if an==`year', d
replace drlsnd=. if drlsnd>`r(p99)' & an==`year'
replace drlsnd=. if drlsnd<`r(p1)' & an==`year'
}
drop if missing(drlsnd)
drop if missing(basman)
drop if missing(bascom)

/* weight inverse of the size of the group in the CZ */
save temp.dta, replace
gen dummy=1
collapse (sum) nbze = dummy , by(ze an)
gen iw2=1/ nbze

save iw2.dta, replace
use temp.dta, replace
capture drop _merge
joinby ze an using iw2.dta , unm(m)
drop _merge

save temp1.dta, replace
collapse r`outc' dimm dimm75 Y_* REG_* , by(ze an)

/* OLS */
ivreg2 r`outc' dimm Y_*  REG_* , partial(Y_* REG_* ) cl(ze)
estimates store `occp'ols

/* IV */
ivreg2 r`outc' (dimm  = dimm75 ) Y_* REG_* , partial(Y_* REG_* ) cl(ze)
estimates store `occp'iv

use temp1.dta, clear
ivreg2 r`outc' (dimm inter = dimm75 inter75) lqinit  Y_*  REG_* [aweight = iw2], partial(Y_* REG_* ) cl(ze)
estimates store `occp'ivs

test dimm + inter = 0
end

capture program drop mobtyp
program define mobtyp
version 11
syntax, outc(name) file(name) 

/* OUTFLOWS */
regsamp, occp(all) outc(`outc')
regsamp, occp(cad) outc(`outc')
regsamp, occp(tecemp2) outc(`outc')
regsamp, occp(bc) outc(`outc')
regsamp, occp(bcnq) outc(`outc')

capture rm "C:\Users\Public\Documents\Tab21\\`file'.rtf"
/* OLS */
esttab allols  cadols tecemp2ols bcols bcnqols /* bcols tradols ntradols constbcols */ ///
using "C:\Users\Public\Documents\Tab21\\`file'.rtf" /// 
, append title("s")  b(%9.3f) cells(b(star fmt(3)) se(fmt(3) par) ) ///
 stats(N widstat) star(* 0.10 ** 0.05 *** 0.01)

/* IV */
esttab alliv  cadiv tecemp2iv bciv bcnqiv /* bciv tradiv ntradiv constbciv */ ///
using "C:\Users\Public\Documents\Tab21\\`file'.rtf" /// 
, append title("s")  b(%9.3f) cells(b(star fmt(3)) se(fmt(3) par) ) ///
 stats(N widstat) star(* 0.10 ** 0.05 *** 0.01)

/* IVSELECT */
esttab allivs  cadivs tecemp2ivs bcivs bcnqivs /* bcivs tradivs ntradivs constbcivs */ ///
using "C:\Users\Public\Documents\Tab21\\`file'.rtf" /// 
, append title("s")  b(%9.3f) cells(b(star fmt(3)) se(fmt(3) par) ) ///
 stats(N widstat) star(* 0.10 ** 0.05 *** 0.01)

end

mobtyp , outc(lshift) file(TableA11_dep_outflows)
